
7 Direct Methods

Direct methods rely solely on the objective function f . These methods are also
called zero-order, black box, pattern search, or derivative-freemethods. Direct methods
do not rely on derivative information to guide them toward a local minimum
or identify when they have reached a local minimum. They use other criteria to
choose the next search direction and to judge when they have converged.

7.1 Cyclic Coordinate Search x1

2

Figure 7.1. Cyclic coordinate de-
scent alternates between coordi-
nate directions.

x

Figure 7.2. Above is an example of
how cyclic coordinate search can
get stuck. Moving in either of the
coordinate directions will result
only in increasing f , but moving
diagonally, which is not allowed in
cyclic coordinate search, can result
in lowering f .

Cyclic coordinate search, also known as coordinate descent or taxicab search, simply
alternates between coordinate directions for its line search. The search starts from
an initial x(1) and optimizes the first input:

Having solved this, it optimizes the next coordinate:

n ) (7.2)

This process is equivalent to doing a sequence of line searches along the set of n

basis vectors, where the ith basis vector is all zero except for the ith component,
which has value 1 (algorithm 7.1). For example, the third basis function, denoted
e(3), in a four-dimensional space is:

e(3) = [0, 0, 1, 0] (7.3)

basis(i, n) = [k == i ? 1.0 : 0.0 for k in 1 : n] Algorithm 7.1. A function for con-
structing the ith basis vector of
length n.

Figure 7.1 shows an example of a search through a two-dimensional space.
Like steepest descent, cyclic coordinate search is guaranteed either to improve

or to remain the same with each iteration. No significant improvement after a full
cycle over all coordinates indicates that the method has converged. Algorithm 7.2
provides an implementation. As figure 7.2 shows, cyclic coordinate search can
fail to find even a local minimum.

function cyclic_coordinate_descent(f, x, ϵ)
Δ, n = Inf, length(x)
while abs(Δ) > ϵ

x′ = copy(x)
for i in 1 : n

d = basis(i, n)
x = line_search(f, x, d)

end
Δ = norm(x - x′)

end
return x

end

Algorithm 7.2. The cyclic coordi-
nate descent method takes as in-
put the objective function f and a
starting point x, and it runs until
the step size over a full cycle is less
than a given tolerance ϵ.

The method can be augmented with an acceleration step to help traverse
diagonal valleys. For every full cycle starting with optimizing x(1) along e(1)

and ending with x(n+1) after optimizing along e(n), an additional line search is
conducted along the direction x(n+1) − x(1). An implementation is provided in
algorithm 7.3 and an example search trajectory is shown in figure 7.3.

Figure 7.3. Adding the accelera-
tion step to cyclic coordinate de-
scent helps traverse valleys. Six
steps are shown for both the origi-
nal and accelerated versions.

7.2 Powell’s Method

Powell’s method1 can search in directions that are not orthogonal to each other. The

1 Powell’s method was first intro-
duced by M. J.D. Powell, “An Ef-
ficient Method for Finding the
Minimum of a Function of Sev-
eral Variables Without Calculat-
ing Derivatives,” Computer Journal,
vol. 7, no. 2, pp. 155–162, 1964.
An overview is presented by W.H.
Press, S.A. Teukolsky, W. T. Vetter-
ling, and B. P. Flannery, Numerical
Recipes in C: The Art of Scientific
Computing. Cambridge University
Press, 1982, vol. 2.

method can automatically adjust for long, narrow valleys that might otherwise
require a large number of iterations for cyclic coordinate descent or other methods
that search in axis-aligned directions.

function cyclic_coordinate_descent_with_acceleration_step(f, x, ϵ)
Δ, n = Inf, length(x)
while abs(Δ) > ϵ

x′ = copy(x)
for i in 1 : n

d = basis(i, n)
x = line_search(f, x, d)

end
x = line_search(f, x, x - x′) # acceleration step
Δ = norm(x - x′)

end
return x

end

Algorithm 7.3. The cyclic coordi-
nate descent method with an ac-
celeration step takes as input the
objective function f and a starting
point x, and it runs until the step
size over a full cycle is less than a
given tolerance ϵ.

x1

x
2

Figure 7.4. Powell’s method starts
the same as cyclic coordinate de-
scent but iteratively learns conju-
gate directions.

The algorithm maintains a list of search directions u(1), . . . , u(n), which are
initially the coordinate basis vectors, u(i) = e(i) for all i. Starting at x(1), Powell’s
method conducts a line search for each search direction in succession, updating
the design point each time:

x(i+1) ← line_search( f , x(i), u(i)) for all i in {1, . . . , n} (7.4)

Next, all search directions are shifted down by one index, dropping the oldest
search direction, u(1):

u(i) ← u(i+1) for all i in {1, . . . , n− 1} (7.5)

The last search direction is replaced with the direction from x(1) to x(n+1), which
is the overall direction of progress over the last cycle:

u(n) ← x(n+1) − x(1) (7.6)

and another line search is conducted along the new direction to obtain a new x(1).
This process is repeated until convergence. Algorithm 7.4 provides an implemen-
tation. Figure 7.4 shows an example search trajectory.

Powell showed that for quadratic functions, after k full iterations the last k

directions will be mutually conjugate. Recall that n line searches along mutually
conjugate directions will optimize an n-dimensional quadratic function. Thus, n

full iterations of Powell’s method, totaling n(n + 1) line searches, will minimize a
quadratic function.

function powell(f, x, ϵ)
n = length(x)
U = [basis(i,n) for i in 1 : n]
Δ = Inf
while Δ > ϵ

x′ = x
for i in 1 : n

d = U[i]
x′ = line_search(f, x′, d)

end
for i in 1 : n-1

U[i] = U[i+1]
end
U[n] = d = x′ - x
x′ = line_search(f, x′, d)
Δ = norm(x′ - x)
x = x′

end
return x

end

Algorithm 7.4. Powell’s method,
which takes the objective function
f, a starting point x, and a tolerance
ϵ.

The procedure of dropping the oldest search direction in favor of the overall
direction of progress can lead the search directions to become linearly dependent.
Without search vectors that are linearly independent, the search directions can
no longer cover the full design space, and the method may not be able to find the
minimum. This weakness can be mitigated by periodically resetting the search
directions to the basis vectors. One recommendation is to reset every n or n + 1

iterations.

7.3 Hooke-Jeeves

TheHooke-Jeeves method (algorithm 7.5) traverses the search space based on evalu-
ations at small steps in each coordinate direction.2 At every iteration, the Hooke- 2 R. Hooke and T.A. Jeeves, “Direct

Search Solution of Numerical and
Statistical Problems,” Journal of the
ACM (JACM), vol. 8, no. 2, pp. 212–
229, 1961.

Jeeves method evaluates f (x) and f (x± αe(i)) for a given step size α in every
coordinate direction from an anchoring point x. It accepts any improvement it may
find. If no improvements are found, it will decrease the step size. The process
repeats until the step size is sufficiently small. Figure 7.5 shows a few iterations
of the algorithm.

Figure 7.5. The Hooke-Jeeves
method, proceeding left to right.
It begins with a large step size
but then reduces it once it cannot
improve by taking a step in any
coordinate direction.

One step of the Hooke-Jeeves method requires 2n function evaluations for
an n-dimensional problem, which can be expensive for problems with many
dimensions. TheHooke-Jeevesmethod is susceptible to local minima. Themethod
has been proven to converge on certain classes of functions.3 3 E.D. Dolan, R.M. Lewis, and V.

Torczon, “On the Local Conver-
gence of Pattern Search,” SIAM
Journal on Optimization, vol. 14,
no. 2, pp. 567–583, 2003.

7.4 Generalized Pattern Search

In contrast with the Hooke-Jeeves method, which searches in the coordinate
directions, generalized pattern search can search in arbitrary directions. A pattern P
can be constructed from a set of directions D about an anchoring point x with a
step size α according to:

P = {x + αd for each d in D} (7.7)

The Hooke-Jeeves method uses 2n directions for problems in n dimensions, but
generalized pattern search can use as few as n + 1.

For generalized pattern search to converge to a local minimum, certain con-
ditions must be met. The set of directions must be a positive spanning set, which
means that we can construct any point in Rn using a nonnegative linear combina-
tion of the directions in D. A positive spanning set ensures that at least one of the
directions is a descent direction from a location with a nonzero gradient.4

4 Convergence guarantees for gen-
eralized pattern search require that
all sampled points fall on a scaled
lattice. Each direction must thus be
a product d(j) = Gz(j) for a fixed
nonsingular n × n matrix G and
integer vector z. V. Torczon, “On
the Convergence of Pattern Search
Algorithms,” SIAM Journal of Op-
timization, vol. 7, no. 1, pp. 1–25,
1997.

We can determine whether a given set of directions D = {d(1), d(2), . . . , d(m)}
in Rn is a positive spanning set. First, we construct the matrix D whose columns
are the directions in D (see figure 7.6). The set of directions D is a positive span-
ning set if D has full row rank and if Dx = −D1 with x ≥ 0 has a solution.5 This 5 R.G. Regis, “On the Properties of

Positive Spanning Sets and Positive
Bases,” Optimization and Engineer-
ing, vol. 17, no. 1, pp. 229–262, 2016.

optimization problem is identical to the initialization phase of a linear program,
which is covered in chapter 11.

function hooke_jeeves(f, x, α, ϵ, γ=0.5)
y, n = f(x), length(x)
while α > ϵ

improved = false
x_best, y_best = x, y
for i in 1 : n

for sgn in (-1,1)
x′ = x + sgn*α*basis(i, n)
y′ = f(x′)
if y′ < y_best

x_best, y_best, improved = x′, y′, true
end

end
end
x, y = x_best, y_best

if !improved
α *= γ

end
end
return x

end

Algorithm 7.5. The Hooke-Jeeves
method, which takes the target
function f, a starting point x, a start-
ing step size α, a tolerance ϵ, and
a step decay γ. The method runs
until the step size is less than ϵ and
the points sampled along the coor-
dinate directions do not provide an
improvement. Based on the imple-
mentation from A. F. Kaupe Jr, “Al-
gorithm 178: Direct Search,” Com-
munications of the ACM, vol. 6, no. 6,
pp. 313–314, 1963.

only positively spans the
cone

only positively spans 1d
space

positively spans R2

Figure 7.6. A valid pattern for gen-
eralized pattern search requires a
positive spanning set. These direc-
tions are stored in the set D.

x(1) x(2) = x(3)

x(4) = x(5)

Figure 7.7. All previous points in
generalized pattern search lie on a
scaled lattice, ormesh. The lattice is
not explicitly constructed and need
not be axis-aligned.

The implementation of generalized pattern search in algorithm 7.6 contains
additional enhancements over the original Hooke-Jeeves method.6 First, the im- 6 These enhancements are pre-

sented in C. Audet and J. E. Dennis
Jr., “Mesh Adaptive Direct Search
Algorithms for Constrained Opti-
mization,” SIAM Journal on Opti-
mization, vol. 17, no. 1, pp. 188–217,
2006.

plementation is opportunistic—as soon as an evaluation improves the current best
design, it is accepted as the anchoring design point for the next iteration. Second,
the implementation uses dynamic ordering to accelerate convergence—a direction
that leads to an improvement is promoted to the beginning of the list of directions.
Figure 7.7 shows a few iterations of the algorithm.

7.5 Nelder-Mead Simplex Method

The Nelder-Mead simplex method7 uses a simplex to traverse the space in search 7 The original simplex method
is covered in J.A. Nelder and
R. Mead, “A Simplex Method
for Function Minimization,” The
Computer Journal, vol. 7, no. 4,
pp. 308–313, 1965. We incorpo-
rate the improvement in J. C. La-
garias, J. A. Reeds, M.H. Wright,
and P. E. Wright, “Convergence
Properties of the Nelder–Mead
Simplex Method in Low Dimen-
sions,” SIAM Journal on Optimiza-
tion, vol. 9, no. 1, pp. 112–147, 1998.

of a minimum. A simplex is a generalization of a tetrahedron to n-dimensional
space. A simplex in one dimension is a line, and in two dimensions it is a triangle
(see figure 7.8). The simplex derives its name from the fact that it is the simplest
possible polytope in any given space.

function generalized_pattern_search(f, x, α, D, ϵ, γ=0.5)
y, n = f(x), length(x)
while α > ϵ

improved = false
for (i,d) in enumerate(D)

x′ = x + α*d
y′ = f(x′)
if y′ < y

x, y, improved = x′, y′, true
D = pushfirst!(deleteat!(D, i), d)
break

end
end
if !improved

α *= γ
end

end
return x

end

Algorithm 7.6. Generalized pat-
tern search, which takes the tar-
get function f, a starting point x,
a starting step size α, a set of search
directions D, a tolerance ϵ, and a
step decay γ. The method runs un-
til the step size is less than ϵ and
the points sampled along the coor-
dinate directions do not provide an
improvement.

Figure 7.8. A simplex in two di-
mensions is a triangle. In order for
the simplex to be valid, it must
have a nonzero area.

The Nelder-Mead method uses a series of rules that dictate how the simplex is
updated based on evaluations of the objective function at its vertices. A flowchart
outlines the procedure in figure 7.9, and algorithm 7.7 provides an implementa-
tion. Like the Hooke-Jeeves method, the simplex can move around while roughly
maintaining its size, and it can shrink down as it approaches an optimum.

The simplex consist of the points x(1), . . . , x(n+1). Let xh be the vertex with the
highest function value, let xs be the vertex with the second highest function value,
and let xℓ be the vertex with the lowest function value. Let x̄ be the mean of all
vertices except the highest point xh. Finally, for any design point xθ , let yθ = f (xθ)

be its objective function value. A single iteration then evaluates four simplex
operations:

Reflection. xr = x̄+ α(x̄− xh), reflects the highest-valued point over the centroid.
This typically moves the simplex from high regions toward lower regions. Here,
α > 0 and is typically set to 1.

Initial simplex

Sort the simplex entries
Compute x̄

Compute the reflection point:
xr = x̄ + α(x̄− xh)

yr < yl?

yes

Compute the expansion point:
xe = x̄ + β(xr − x̄)

ye < yr?

yes

no

Replace xh with xe

no yr ≥ ys?

no

Replace xh with xr

Converged? yes Return best point

no

yes yr ≥ yh? no

Replace xh with xryes

Compute the contraction point:
xc = x̄ + γ(xh − x̄)

yc > yh?

no

yes

Replace xh with xc

Shrink by replacing all
x(i) with (x(i) + xℓ)/2

Figure 7.9. Flowchart for the
Nelder-Mead algorithm.

Expansion. xe = x̄ + β(xr − x̄), like reflection, but the reflected point is sent even
further. This is done when the reflected point has an objective function value
less than all points in the simplex. Here, β > max(1, α) and is typically set to 2.

Contraction. xc = x̄ + γ(xh − x̄), the simplex is shrunk down by moving away
from the worst point. It is parameterized by γ ∈ (0, 1) which is typically set to
0.5.

Shrinkage. All points are moved toward the best point, typically halving the
separation distance.

Figure 7.10 shows the four simplex operations. Figure 7.11 shows several iterations
of the algorithm.

xh

xℓ

xs

x̄

reflection

xr

xh

xℓ

xs

x̄ xr

expansion

xe

xh

xℓ

xs

x̄

contraction

xc
xh

xℓ

xs

shrinkage
Figure 7.10. The Nelder-Mead sim-
plex operations visualized in two-
dimensions.The convergence criterion for the Nelder-Mead simplex method is unlike

Powell’s method in that it considers the variation in the function values rather
than the changes to the points in the design space. It compares the standard
deviation of the sample8 y(1), . . . , y(n+1) to a tolerance ǫ. This value is high for a 8 The standard deviation of the sample

is also called the uncorrected sample
standard deviation. In our case, it is
√

1
n+1 ∑

n+1
i=1

(
y(i) − ȳ

)2, where ȳ is
the mean of y(1), . . . , y(n+1).

simplex over a highly curved region, and it is low for a simplex over a flat region.
A highly curved region indicates that there is still further optimization possible.

7.6 Divided Rectangles

The divided rectangles algorithm, or DIRECT for DIvided RECTangles, is a Lip-
schitzian optimization approach, similar in some ways to the Shubert-Piyavskii
method described in section 3.6.9 However, it eliminates the need for specifying a

9 D.R. Jones, C.D. Perttunen, and
B. E. Stuckman, “Lipschitzian Op-
timization Without the Lipschitz
Constant,” Journal of Optimization
Theory and Application, vol. 79, no. 1,
pp. 157–181, 1993.

Lipschitz constant, and it can be more efficiently extended to multiple dimensions.

Figure 7.11. The Nelder-Mead
method, proceeding left to right
and top to bottom.

function nelder_mead(f, S, ϵ; α=1.0, β=2.0, γ=0.5)
Δ, y_arr = Inf, f.(S)
while Δ > ϵ

p = sortperm(y_arr) # sort lowest to highest
S, y_arr = S[p], y_arr[p]
xl, yl = S[1], y_arr[1] # lowest
xh, yh = S[end], y_arr[end] # highest
xs, ys = S[end-1], y_arr[end-1] # second-highest
xm = mean(S[1:end-1]) # centroid
xr = xm + α*(xm - xh) # reflection point
yr = f(xr)

if yr < yl
xe = xm + β*(xr-xm) # expansion point
ye = f(xe)
S[end],y_arr[end] = ye < yr ? (xe, ye) : (xr, yr)

elseif yr ≥ ys
if yr < yh

xh, yh, S[end], y_arr[end] = xr, yr, xr, yr
end
xc = xm + γ*(xh - xm) # contraction point
yc = f(xc)
if yc > yh

for i in 2 : length(y_arr)
S[i] = (S[i] + xl)/2
y_arr[i] = f(S[i])

end
else

S[end], y_arr[end] = xc, yc
end

else
S[end], y_arr[end] = xr, yr

end

Δ = std(y_arr, corrected=false)
end
return S[argmin(y_arr)]

end

Algorithm 7.7. The Nelder-Mead
simplex method, which takes the
objective function f, a starting sim-
plex S consisting of a list of vec-
tors, and a tolerance ϵ. The Nelder-
Mead parameters can be speci-
fied as well and default to recom-
mended values.

The notion of Lipschitz continuity can be extended to multiple dimensions. If
f is Lipschitz continuous over a domain X with Lipschitz constant ℓ > 0, then
for a given design x(1) and y = f (x(1)), the circular cone

f (x(1))− ℓ‖x− x(1)‖2 (7.8)

forms a lower bound of f . Given m function evaluations with design points
{x(1), . . . , x(m)}, we can construct a superposition of these lower bounds by taking
their maximum:

maximize
i

f (x(i))− ℓ‖x− x(i)‖2 (7.9)

−2 −1 0 1 2
−2

−1

0

1

2

x1

x
2

Lipschitz lower bound

−2 −1 0 1 2

x1

divided rectangles lower bound Figure 7.12. The Lipschitz lower
bound is an intersection of cones,
which creates complicated surfaces
in multidimensional space. The di-
vided rectangle lower bound iso-
lates each lower-bound cone to
its own hyper-rectangular region,
making it trivial to compute the
minimum value in each region
given a Lipschitz constant.

The Shubert-Piyavskii method samples at a point where the bound derived
from a known Lipschitz constant is lowest. Unfortunately, a Lipschitz lower bound
has intricate geometry whose complexity increases with the dimensionality of
the design space. The left contour plot in figure 7.12 shows such a lower bound
using five function evaluations. The right contour plot shows the approximation
made by DIRECT, which divides the region into hyper-rectangles—one centered
about each design point. Making this assumption allows for the rapid calculation
of the minimum of the lower bound.

The DIRECT method does not assume a Lipschitz constant is known. Fig-
ures 7.13 and 7.14 show lower bounds constructed using Lipschitz continuity
and the DIRECT approximation, respectively, for several different Lipschitz con-
stants. Notice that the location of the minimum changes as the Lipschitz constant
changes, with small values of ℓ leading to designs near the lowest function evalua-
tions and larger values of ℓ leading to designs the farthest from previous function
evaluations.

x

y

f (x)

ℓ = 0.1

ℓ = 0.2

ℓ = 0.3

ℓ = 0.4

ℓ = 0.5

ℓ = 0.6

ℓ = 0.7

ℓ = 0.8

ℓ = 0.9

ℓ = 1.0

lower-bound minima
evaluation points

Figure 7.13. The Lipschitz lower
bound for different Lipschitz con-
stants ℓ. Not only does the esti-
mated minimum change locally as
the Lipschitz constant is varied, the
region in which the minimum lies
can vary as well.

ℓ = 0.9

ℓ = 1.0

lower-bound minima
evaluation points

Figure 7.14. The DIRECT lower
bound for different Lipschitz con-
stants ℓ. The lower bound is not
continuous. The minimum does
not change locally but can change
regionally as the Lipschitz constant
changes.

7.6.1 Univariate DIRECT
In one dimension, DIRECT recursively divides intervals into thirds and then
samples the objective function at the center of the intervals as shown in figure 7.15.
This scheme is in contrast with the Shubert-Piyavskii method, where sampling
occurs at a point where the bound derived from a known Lipschitz constant is
lowest.

before division:
after division:

Figure 7.15. Center-point sampling,
using the DIRECT scheme, divides
intervals into thirds.

For an interval [a, b] with center c = (a + b)/2, the lower bound based on f (c)

is:
f (x) ≥ f (c)− ℓ|x− c| (7.10)

where ℓ is a Lipschitz constant that is unknown to us. The lowest value that the
bound obtains on that interval is f (c)− ℓ(b− a)/2, and it occurs at the edges of
the interval.

Even though we do not know ℓ, we can deduce that the lower bound of some
intervals are lower than others. If we have two intervals of the same length, for
example, and the first one has a lower evaluation at the center point than the
second one, then the lower bound of the first interval is lower than that of the
second. Although this does not entail that the first interval contains a minimizer,
it is an indication that we may want to focus our search in that interval.

During our search, we will have many intervals [a1, b1], . . . , [an, bn] of different
widths. We can plot our intervals according to their center value and interval
width, as we do in figure 7.16. The lower bound for each interval is the vertical
intercept of a line of slope ℓ through its center point. The center of the interval
with the lowest lower bound will be the first point intersected as we shift a line
with slope ℓ upwards from below.

DIRECT splits all intervals for which a Lipschitz constant exists such that
they have the lowest lower bound, as shown in figure 7.17. We refer to these
selected intervals as potentially optimal. Any interval may technically contain
the optimum, though the selected points heuristically have the best chance of
containing the optimum.

One iteration of the one-dimensional DIRECT algorithm consists of identifying
the set of potentially optimal intervals and then dividing each interval into thirds.
Example 7.1 demonstrates the univariate DIRECT method.

f (ci)− ℓ(bi − ai)/2

f (ci)

(bi − ai)/2

slope = ℓ

Figure 7.16. Interval selection for
a particular Lipschitz constant ℓ.
Black dots represent DIRECT inter-
vals and the corresponding func-
tion evaluations at their centers.
A black line of slope ℓ is drawn
through the dot belonging to the se-
lected interval. All other dots must
lie on or above this line.

suboptimal

Figure 7.17. The potentially op-
timal intervals for the DIRECT
method form a piecewise bound-
ary that encloses all intervals along
the lower-right. Each dot corre-
sponds to an interval.

Consider the function f (x) = sin(x) + sin(2x) + sin(4x) + sin(8x) on the
interval [−2, 2]with a global minimizer near−0.272. Optimization is difficult
because of multiple local minima.

The figure below shows the progression of the univariate DIRECTmethod,
with intervals chosen for splitting rendered in blue. The left side shows the
intervals overlaid on the objective function. The right side shows the intervals
as scatter points in the interval half-width versus center value space.

Example 7.1. The DIRECT method
applied to a univariate function.

7.6.2 Multivariate DIRECT
In multiple dimensions, we divide rectangles (or hyper-rectangles in more than
two dimensions) instead of intervals. Similar to the univariate case, we divide the
rectangles into thirds along the axis directions. Before commencing the division
of the rectangles, DIRECT normalizes the search space to be the unit hypercube.

As illustrated in figure 7.18, the choice of ordering of the directions when
splitting the unit hypercube matters. DIRECT prioritizes the assignment of larger
rectangles for the points with lower function evaluations. Larger rectangles are
prioritized for additional splitting.

sampling

horizontal split vertical split

vertical split horizontal split

Figure 7.18. Interval splitting in
multiple dimensions (using DI-
RECT) requires choosing an order-
ing for the split dimensions.

When splitting a regionwithout equal side lengths, only the longest dimensions
are split (figure 7.19). Splitting then proceeds on these dimensions in the same
manner as with a hypercube.

Figure 7.19. DIRECTwill split only
the longest dimensions of hyper-
rectangles.

The set of potentially optimal intervals is obtained as it is with one dimension.
The lower bound for each hyper-rectangle can be computed based on the longest
center-to-vertex distance and center value. We can construct a diagram similar to
figure 7.16 to identify the potentially optimal rectangles.10 10 As an additional requirement for

selection, DIRECT also requires
that the lower bound for the inter-
val improve the current best value
by a nontrivial amount.

7.6.3 Implementation
DIRECT (algorithm 7.8) is best understood when it is broken down into subrou-
tines. We present these subroutines below.

Normalization to the unit hypercube is done by algorithm 7.9.

function direct(f, a, b, ϵ, k_max)
g = reparameterize_to_unit_hypercube(f, a, b)
intervals = Intervals()
n = length(a)
c = fill(0.5, n)
interval = Interval(c, g(c), fill(0, n))
add_interval!(intervals, interval)
c_best, y_best = copy(interval.c), interval.y

for k in 1 : k_max
S = get_opt_intervals(intervals, ϵ, y_best)
to_add = Interval[]
for interval in S

append!(to_add, divide(g, interval))
dequeue!(intervals[vertex_dist(interval)])

end
for interval in to_add

add_interval!(intervals, interval)
if interval.y < y_best

c_best, y_best = copy(interval.c), interval.y
end

end
end

return rev_unit_hypercube_parameterization(c_best, a, b)
end

Algorithm 7.8. DIRECT, which
takes the multidimensional objec-
tive function f, vector of lower
bounds a, vector of upper bounds
b, tolerance parameter ϵ, and num-
ber of iterations k_max. It returns
the best coordinate.

rev_unit_hypercube_parameterization(x, a, b) = x.*(b-a) + a
function reparameterize_to_unit_hypercube(f, a, b)

Δ = b-a
return x->f(x.*Δ + a)

end

Algorithm 7.9. A function that cre-
ates a function defined over the
unit hypercube that is a reparam-
eterized version of the function f
defined over the hypercube with
lower and upper bounds a and b.

Computing the set of potentially optimal rectangles can be done efficiently
(algorithm 7.10). We use the fact that the interval width can only take on powers
of one-third, and thus many of the points will share the same x coordinate. For
any given x coordinate, only the one with the lowest y value can be a potentially
optimal interval. We store the rectangular intervals according to their depth and
then in a priority queue according to their centerpoint value.

using DataStructures
struct Interval

c
y
depths

end
min_depth(interval) = minimum(interval.depths)
vertex_dist(interval) = norm(0.5*3.0.^(-interval.depths), 2)
const Intervals =

OrderedDict{Float64,PriorityQueue{Interval, Float64}}
function add_interval!(intervals, interval)

d = vertex_dist(interval)
if !haskey(intervals, d)

intervals[d] = PriorityQueue{Interval, Float64}()
end
return enqueue!(intervals[d], interval, interval.y)

end

Algorithm 7.10. The data structure
used in DIRECT. Here, Interval
has three fields: the interval center
c, the center point value y = f(c),
and the number of divisions in
each dimension depths. The func-
tion add_interval! inserts a new
Interval into the data structure.

We can use this data structure to obtain all potentially optimal intervals (algo-
rithm 7.11). The algorithm proceeds from lowest interval width to highest interval
width. For each point, we first determine whether it is above or below the line
joining the previous two points. If it is below, we skip it. The same determination
is then made for the next point.

Finally, we need a method for dividing the intervals. This is implemented by
the divide method (algorithm 7.12).

The intervals obtained from running DIRECT in two dimensions are visualized
in figure 7.20. Two iterations of DIRECT in two dimensions are worked out in
example 7.2.

function get_opt_intervals(intervals, ϵ, y_best)
stack = Interval[]
for (x, pq) in intervals

if !isempty(pq)
interval = DataStructures.peek(pq)[1]
y = interval.y

while length(stack) > 1
interval1 = stack[end]
interval2 = stack[end-1]
x1, y1 = vertex_dist(interval1), interval1.y
x2, y2 = vertex_dist(interval2), interval2.y
ℓ = (y2 - y) / (x2 - x)
if y1 <= ℓ*(x1-x) + y + ϵ

break
end
# remove previous interval
pop!(stack)

end

if !isempty(stack) && interval.y > stack[end].y + ϵ
# skip new interval
continue

end

push!(stack, interval) # add new interval
end

end
return stack

end

Algorithm 7.11. A routine for ob-
taining the potentially optimal in-
tervals, where intervals is of type
Intervals, ϵ is a tolerance parame-
ter, and y_best is the best function
evaluation.

function divide(f, interval)
c, d, n = interval.c, min_depth(interval), length(interval.c)
dirs = findall(interval.depths .== d)
cs = [(c + 3.0^(-d-1)*basis(i,n),

c - 3.0^(-d-1)*basis(i,n)) for i in dirs]
vs = [(f(C[1]), f(C[2])) for C in cs]
minvals = [min(V[1], V[2]) for V in vs]

intervals = Interval[]
depths = copy(interval.depths)
for j in sortperm(minvals)

depths[dirs[j]] += 1
C, V = cs[j], vs[j]
push!(intervals, Interval(C[1], V[1], copy(depths)))
push!(intervals, Interval(C[2], V[2], copy(depths)))

end
push!(intervals, Interval(c, interval.y, copy(depths)))
return intervals

end

Algorithm 7.12. The divide rou-
tine for dividing an interval, where
f is the objective function and
interval is the interval to be di-
vided. It returns a list of the result-
ing smaller intervals.

7.7 Summary

• Direct methods rely solely on the objective function and do not use derivative
information.

• Cyclic coordinate search optimizes one coordinate direction at a time.

• Powell’s method adapts the set of search directions based on the direction of
progress.

• Hooke-Jeeves searches in each coordinate direction from the current point
using a step size that is adapted over time.

• Generalized pattern search is similar to Hooke-Jeeves, but it uses fewer search
directions that positively span the design space.

• The Nelder-Mead simplex method uses a simplex to search the design space,
adaptively expanding and contracting the size of the simplex in response to
evaluations of the objective function.

• The divided rectangles algorithm extends the Shubert-Piyavskii approach to
multiple dimensions and does not require specifying a valid Lipschitz constant.

Figure 7.20. The DIRECT method
after 16 iterations on the Branin
function, appendix B.3. Each cell is
bordered by white lines. The cells
are much denser around the min-
ima of the Branin function, as the
DIRECT method procedurally in-
creases its resolution in those re-
gions.

Consider using DIRECT to optimize the flower function (appendix B.4)
over x1 ∈ [−1, 3], x2 ∈ [−2, 1]. The function is first normalized to the unit
hypercube such that we optimize x′1, x′2 ∈ [0, 1]:

f (x′1, x′2) = flower(4x′1 − 1, 3x′2 − 2)

The objective function is sampled at [0.5, 0.5] to obtain 0.158. We have a
single interval with center [0.5, 0.5] and side lengths [1, 1]. The interval is
divided twice, first into thirds in x′1 and then the center interval is divided
into thirds in x′2.

x′10 1

x′2

0

1

We now have five intervals:

interval center side lengths vertex distance center value
1 [1/6, 3/6] [1/3, 1] 0.527 0.500

2 [5/6, 3/6] [1/3, 1] 0.527 1.231

3 [3/6, 3/6] [1/3, 1/3] 0.236 0.158

4 [3/6, 1/6] [1/3, 1/3] 0.236 2.029

5 [3/6, 5/6] [1/3, 1/3] 0.236 1.861

We next split on the two intervals centered at [1/6, 3/6] and [3/6, 3/6].

Example 7.2. The first two itera-
tions of DIRECT worked out in de-
tail.

7.8 Exercises

Exercise 7.1. Previous chapters covered methods that leverage the derivative
to descend toward a minimum. Direct methods are able to use only zero-order
information—evaluations of f . Howmany evaluations are needed to approximate
the derivative and the Hessian of an n-dimensional objective function using finite
differencemethods?Why do you think it is important to have zero-ordermethods?

Exercise 7.2. Design an objective function and a starting point x0 such that Hooke-
Jeeves will fail to attain a reduction in the objective function. You need to choose
x0 such that it is not a local minimizer.

Exercise 7.3. Is the design point obtained using the Hooke-Jeeves method guar-
anteed to be within ǫ of a local minimum?

Exercise 7.4. Give an example of a concrete engineering problem where you may
not be able to compute analytical derivatives.

Exercise 7.5. State a difference between the divided rectangles algorithm in one
dimension and the Shubert-Piyavskii method.

Exercise 7.6. Suppose our search algorithmhas us transition from x(k) = [1, 2, 3, 4]

to x(k+1) = [2, 2, 2, 4]. Could our search algorithm be (a) cyclic coordinate search,
(b) Powell’s method, (c) both a and b, or (d) neither a nor b? Why?

